1 SUMMARY

To **solve one or more sets of variable-band linear equations, $\mathbf{Ax}=\mathbf{b}$ or $\mathbf{A}^{\mathrm{T}}\mathbf{x}=\mathbf{b}$ by the frontal method.**

MA43 provides the user with a straightforward interface to the routine MA42 when entry is by equations and auxiliary storage is not required. If the user requires more sophisticated facilities, MA42 should be employed. It is recommended that **A** is preordered using the HSL routine MC62.

ATTRIBUTES — Version: 1.0.1. (29th November 2022) **Types:** Real (single, double). **Helpful:** MC62. **Calls:** MA42, MC59. **Original date:** March 1993. **Origin:** J.A. Scott, Rutherford Appleton Laboratory.

2 HOW TO USE THE PACKAGE

2.1 Argument lists and calling sequences

There are four entries:

(a) The initialization subroutine MA43I/ID may first be called.

(b) MA43A/AD accepts the matrix **A** and prepares data structures for the factorization.

(c) MA43B/BD factorizes the matrix **A** using the data from MA43A/AD and, optionally, solves $\mathbf{Ax}=\mathbf{b}$.

(d) MA43C/CD uses the factors produced by MA43B/BD to rapidly solve either further systems of the form $\mathbf{Ax}=\mathbf{b}$ or systems of the form $\mathbf{A}^{\mathrm{T}}\mathbf{x}=\mathbf{b}$.

2.1.1 The initialization subroutine

To initialize control parameters the user may make a call of the following form:

The single precision version

```
CALL MA43I(ICNTL,CNTL)
```

The double precision version

```
CALL MA43ID(ICNTL,CNTL)
```

ICNTL is an INTEGER array of length 3 which need not be set by the user. This array is used to hold control parameters. On exit, ICNTL contains default values (see Section 2.2.1. for details).

CNTL is a REAL (DOUBLE PRECISION in the D version) array of length 2 which need not be set by the user. This array is used to hold control parameters. On exit, CNTL contains default values (see Section 2.2.1. for details).

2.1.2 Specification of the matrix and symbolic factorization

The single precision version

```
CALL MA43A(N,NE,MORE,IRN,JCN,A,IW,KEEP,ICNTL,INFO)
```

The double precision version

```
CALL MA43AD(N,NE,MORE,IRN,JCN,A,IW,KEEP,ICNTL,INFO)
```

N is an INTEGER variable which must be set by the user to the order of the matrix. This argument is not altered by the routine. **Restriction:** $\mathtt{N}\geq 1$.

NE is an INTEGER variable which must be set by the user to the number of the entries in the matrix. This argument is not altered by the routine. **Restriction:** $\mathtt{NE}\geq 1$.

MORE is a LOGICAL variable which must be set to .TRUE. if the user wishes to subsequently call MA43C/CD to solve either further systems of the form $\mathbf{Ax}=\mathbf{b}$ or systems of the form $\mathbf{A}^{\mathrm{T}}\mathbf{x}=\mathbf{b}$, and should be set to .FALSE. otherwise. There are substantial savings in the amount of storage required if MORE = .FALSE. This argument is not altered by the routine.

IRN is an INTEGER array of length NE which must be set by the user to contain the row indices of the entries in the matrix. The entries may be in any order. This array is altered by the routine.

JCN is an INTEGER array of length NE. JCN(K) must be set be the user to contain the column index of the entry whose row index is held in IRN(K), K = 1,2,..., NE. This array is altered by the routine and must be passed unchanged to MA43B/BD.

A is a REAL (DOUBLE PRECISION in the D version) array of length NE. A(K) must be set by the user to contain the value of the entry with indices (IRN(K), JCN(K)), K=1,2,...,NE. This array is altered by the routine and must be passed unchanged to MA43B/BD.

IW is an integer array of length N+1 which is used as workspace.

KEEP is an INTEGER array of length 2*N+60 which need not be set by the user. This array must be passed unchanged to MA43B/BD.

ICNTL is an INTEGER array of length 3 which must be set by the user to hold control parameters. Default values may be set by a call to MA43I/ID. Details of the control parameters are given in Section 2.1.1. This argument is not altered by the routine.

INFO is an INTEGER array of length 9 which need not be set by the user. On successful exit, INFO(1) is set to 0. Negative values of INFO(1) indicate a fatal error has been detected and positive values indicate a warning has been issued (see Section 2.3.1). For details of information contained in INFO(I), I=2,...,6 see Section 2.2.2. INFO(I), $\mathtt{I}\geq 7$, is not accessed by the routine.

2.1.3 To factorize A and optionally solve $\mathbf{Ax}=\mathbf{b}$

The single precision version

```
      CALL MA43B(N,NE,JCN,A,NRHS,LX,B,X,LFACT,FACT,LIFACT,IFACT,
     *           KEEP,ICNTL,CNTL,INFO,RINFO)
```

The double precision version

```
      CALL MA43BD(N,NE,JCN,A,NRHS,LX,B,X,LFACT,FACT,LIFACT,IFACT,
     *            KEEP,ICNTL,CNTL,INFO,RINFO)
```

N,NE are INTEGER variables which must be unchanged since the call to MA43A/AD. These arguments are not altered by the routine.

JCN is an INTEGER array of length NE which must be unchanged since the call to MA43A/AD. This argument is not altered by the routine.

A is a REAL (DOUBLE PRECISION in the D version) array of length NE which must be unchanged since the call to MA43A/AD. This argument is not altered by the routine.

NRHS is an INTEGER variable which must be set by the user to the number of right-hand sides. This argument is not altered by the routine. **Restriction:** $\mathtt{NRHS}\geq 0$.

LX is an INTEGER variable which must be set by the user to the first dimension of arrays B and X. This argument is not altered by the routine. **Restriction:** If $\mathtt{NRHS}\geq 1$, $\mathtt{LX}\geq \mathtt{N}$.

B is a REAL (DOUBLE PRECISION in the D version) array of dimensions LX by max(1, NRHS). B(I,J) must be set by the user to be the I-th component of the J-th right-hand side (I=1,2,...,N, J=1,2,...,NRHS). This argument is not altered by the routine. If $\mathtt{NRHS}=0$, the array B is not accessed.

X is a REAL (DOUBLE PRECISION in the D version) array of dimension LX by max(1, NRHS) which need not be set by the user. On exit, X(I,J) holds the solution for variable I to system J (I=1,2,...,N, J=1,2,..., NRHS). If NRHS = 0, the array X is not accessed.

LFACT is an INTEGER variable which must be set by the user to the dimension of array FACT. LFACT must be at least INFO(2) + INFO(3)*max(1, NRHS) (INFO(2) and INFO(3) as output from MA43A/AD). If **A** is nonsingular, this value is sufficient. If **A** is singular, a larger value may be needed (see Section 2.2.2). This argument is not altered by the routine.

FACT is a REAL (DOUBLE PRECISION in the D version) array of length LFACT which is used to hold the factors of **A**. MA43B/BD. This array must be passed unchanged to any subsequent calls to MA43C/CD.

LIFACT is an INTEGER variable which must be set by the user to the dimension of array IFACT. LIFACT must be at least INFO(4) (as output from MA43A/AD). If **A** is nonsingular, this value is sufficient. If **A** is singular, a larger value may be needed (see Section 2.2.2). This argument is not altered by the routine.

IFACT is an INTEGER array of length LIFACT which is used to hold integer data for the factors of **A**. This array must be passed unchanged to any subsequent calls to MA43C/CD.

KEEP is an INTEGER array of length 2*N+60 which must be unchanged since the call to MA43A/AD. This array is changed by the routine and must be passed without further change to any subsequent calls to MA43C/CD.

ICNTL is an INTEGER array of length 3 which must be set by the user to hold control parameters. Default values may be set by a call to MA43I/ID. Details of the control parameters are given in Section 2.2.1. This argument is not altered by the routine.

CNTL is a REAL (DOUBLE PRECISION in the D version) array of length 2 which must be set by the user to hold control parameters. Default values may be set by a call to MA43I/ID. Details of the control parameters are given in Section 2.2.1. This argument is not altered by the routine.

INFO is an INTEGER array of length 9. On successful exit, INFO(1) is set to 0. Negative values indicate a fatal error. Values greater than 0 are associated with a warning or non-terminal error. For nonzero values of INFO(1), see Section 2.3.2. For details of the information contained in the other components of INFO, see Section 2.2.2.

RINFO is a REAL (DOUBLE PRECISION in the D version) array of length 2 which need not be set by the user. For details of the information contained in RINFO, see Section 2.2.2.

2.1.4 To solve further systems $\mathbf{Ax}=\mathbf{b}$ or systems $\mathbf{A}^{\mathrm{T}}\mathbf{x}=\mathbf{b}$

The single precision version

```
      CALL MA43C(TRANS,N,NRHS,LX,B,X,LFACT,FACT,LIFACT,IFACT,KEEP,
     *           ICNTL,INFO)
```

The double precision version

```
      CALL MA43CD(TRANS,N,NRHS,LX,B,X,LFACT,FACT,LIFACT,IFACT,KEEP,
     *            ICNTL,INFO)
```

TRANS is a LOGICAL variable which must be set by the user. If TRANS=.TRUE. systems of the form $\mathbf{A}^{\mathrm{T}}\mathbf{x}=\mathbf{b}$ are to be solved and if TRANS=.FALSE. systems of the form $\mathbf{Ax}=\mathbf{b}$ are to be solved.

N is an INTEGER variable which must be unchanged since the call to MA43B/BD. This argument is not altered by the routine.

NRHS is an INTEGER variable which must be set by the user to the number of systems which are to be solved. This argument is not altered by the routine. **Restriction:** NRHS ≥ 1.

LX is an INTEGER variable which must be set by the user to the first dimension of arrays B and X. This argument is not altered by the routine. **Restriction:** LX ≥ N.

B is a REAL (DOUBLE PRECISION in the D version) array of dimensions LX by NRHS. B(I,J) must be set by the user to be the I-th component of the J-th right-hand side (I=1,2,...,N, J=1,2,...,NRHS). This argument is changed by the routine.

X is a REAL (DOUBLE PRECISION in the D version) array of dimension LX by NRHS which need not be set by the user. On exit, X(I,J) holds the solution for variable I to system J (I=1,2,...,N, J=1,2,..., NRHS).

LFACT, FACT, LIFACT, IFACT, and KEEP must all be unchanged since the call to MA43B/BD. These arguments are not altered by the routine.

ICNTL is an INTEGER array of length 3 which must be set by the user to hold control parameters. Default values may be set by a call to MA43I/ID. Details of the control parameters are given in Section 2.2.1. This argument is not altered by the routine.

INFO is an INTEGER array of length 9. On successful exit, INFO(1) is set to 0. Negative values indicate a fatal error (see Section 2.3.3). INFO(I), $I \geq 2$, is not accessed by the routine.

2.2 Arrays for control and information

2.2.1 Control parameters

The elements of the arrays ICNTL and CNTL control the action of MA43A/AD, MA43B/BD, and MA43C/CD. Default values are set by MA43I/ID.

ICNTL(1) is the stream number for error messages and has the default value 6. Printing of error messages is suppressed if $\text{ICNTL}(1) \leq 0$.

ICNTL(2) is the stream number for warning messages and has the default value 6. Printing of warning messages is suppressed if $\text{ICNTL}(2) \leq 0$.

ICNTL(3) has the default value 0. If the matrix is found to be singular during the decomposition and ICNTL(3) is equal to 0, an error flag is set and the computation terminates (see INFO(1) = –7 in Section 2.3). If ICNTL(3) is nonzero, a warning is given, the computation continues and components of the solution vector X corresponding to zero pivots are set equal to zero (see also INFO(9) in Section 2.2.2 and INFO(1) = +2 in Section 2.3).

CNTL(1) has the default value zero. The matrix is declared singular if, during the factorization, the entry of largest absolute value in any column is less than or equal to CNTL(1).

CNTL(2) has the default value 0.1 but is not used by the current version of the code.

2.2.2 Information arrays

The elements of the arrays INFO and RINFO provide information on the action of MA43A/AD, MA43B/BD, and MA43C/CD.

INFO(1) is used as an error and a warning flag. If a call to a routine in the MA43 package is successful, on exit INFO(1) has value 0. A nonzero value of INFO(1) indicates an error has been detected or a warning issued (see Section 2.3). If an error is detected during a call to MA43B/BD, the information contained in INFO(I), $I \geq 7$ and in RINFO may be incomplete.

INFO(2), INFO(3) hold, on exit from MA43A/AD, information to assist the user in choosing an appropriate value for the length LFACT of the array FACT required by MA43B/BD. LFACT must be at least INFO(2) + INFO(3)*max(1, NRHS). This value is sufficient unless the matrix **A** is singular, in which case a larger value may be required. If **A** is singular and ICNTL(3) is nonzero and user does not allow sufficient space to permit a successful factorization, on exit from MA43B/BD with INFO(1) = -5, INFO(2) and INFO(3) hold revised values and the user is again advised to set LFACT to be somewhat larger than INFO(2) + INFO(3)*max(1, NRHS).

INFO(4) holds, on exit from MA43A/AD, information to assist the user in choosing an appropriate value for the length LIFACT of the array IFACT required by MA43B/BD. LIFACT must be at least INFO(4). This value is sufficient unless the matrix **A** is singular, in which case a larger value may be required. If **A** is singular and ICNTL(3) is nonzero and user does not allow sufficient space to permit a successful factorization, on exit from MA43B/BD with INFO(1) = -6, INFO(4) holds a revised value and the user is again advised to set LIFACT to be somewhat larger than INFO(4).

INFO(5) holds, on exit from MA43A/AD, the number of multiple entries in the matrix. Such entries are summed.

INFO(6) holds, on exit from MA43A/AD, the number of entries with out-of-range indices. Such entries are ignored.

INFO(7) is set, on exit from MA43B/BD, to +1 (respectively, –1) if the determinant of the matrix is positive (negative). If the matrix is found to be singular, INFO(7) is set to 0. (See also RINFO(1)).

INFO(8) is not used by the current version of the code (on exit, it is set to zero).

INFO(9) holds, on exit from MA43B/BD with INFO(1) = +2 and ICNTL(3) nonzero, an estimate of the deficiency of the matrix. Otherwise, INFO(9) is set to 0.

RINFO(1) holds the natural logarithm of the modulus of the determinant of the matrix **A** (see also INFO(7)). If the matrix is found to be singular, RINFO(1) is set to zero.

RINFO(2) holds the number of floating-point operations in the innermost loops.

2.3 Error diagnostics

On successful completion, the subroutines in the MA43 package will exit with the parameter INFO(1) set to 0. A negative value for INFO(1) is associated with a fatal error. In each case, if ICNTL(1) is greater than zero, a self-explanatory message is output on unit ICNTL(1) (see Section 2.2.1). Warning messages are associated with a positive value of INFO(1). If ICNTL(2) is greater than zero, a self-explanatory message is output on unit ICNTL(2).

2.3.1 Error diagnostics for MA43A/AD

Possible negative values for INFO(1) are:

–1 N has value less than 1.

–2 NE has value less than 1. This error is also returned if all the entries have out-of-range indices.

–3 A row or column of **A** has no entries.

Possible positive values for INFO(1) are:

+1 Multiple entries have been input, or one or more entries in IRN or JCN are out of range, or both are true. (See INFO(5) and INFO(6) in Section 2.2.2).

2.3.2 Error diagnostics for MA43B/BD

Possible negative values for INFO(1) are:

–4 NRHS has value less than 0.

–5 LFACT is too small. (See INFO(2) and INFO(3)).

–6 LIFACT is too small. (See INFO(4)).

–7 Singularity detected in the matrix during the factorization with the control parameter ICNTL(3) equal to zero (see Section 2.2.1).

–8 NRHS ≥ 1 and the defined first dimension LX of the arrays B and X is less than N.

Possible positive values for INFO(1) are:

+2 Singularity detected in the matrix during the factorization with the control parameter ICNTL(3) nonzero (see

Section 2.2.1). Estimated rank of the matrix is held in INFO(9) (see Section 2.2.2).

2.3.3 Error diagnostics for MA43C/CD

Possible negative values for INFO(1) are:

–4 NRHS has value less than 1.

–8 The defined first dimension LX of the arrays X and B is less than N.

–9 MA43C/CD has been called after calling MA43A/AD with MORE = .FALSE.

3 GENERAL INFORMATION

3.1 Summary of information.

Use of common: None.

Other routines called directly: The HSL routines MA42 and MC59.

Workspace: Workspace is provided by the array IW(N+1) (MA43A/AD entry only).

Input/output: In the event of errors, diagnostic messages are printed. The output streams for these messages are controlled by the variables ICNTL(1) and ICNTL(2) (see Section 2.2.1). Stream ICNTL(1) is used for error messages (INFO(1) <0) and stream ICNTL(2) for warnings (INFO(1) >0).

Restrictions:

N ≥ 1 (MA43A/AD entry only),

NE ≥ 1 (MA43A/AD entry only),

NRHS ≥ 0 (MA43B/BD entry only), NRHS ≥ 1 (MA43C/CD entry only).

If NRHS ≥ 1, LX $\geq$ N (MA43B/BD and MA43C/CD entries).

4 METHOD

MA43 provides an interface to MA42 for assembled matrices (that is, equation entry) in the case when direct access data sets are not required. If the user wishes to store the matrix factors in direct access data sets, MA42 should be employed directly.

MA43A/AD calls MC59A/AD to reorder the input data to hold the entries by rows. It then calls MA42A/AD for each row and then MA42J/JD for each row. MA43B/BD performs the actual factorization and optionally solves $\mathbf{Ax}=\mathbf{b}$ by calling MA42B/BD for each row. MA43C/CD solves further systems or transpose systems by calling MA42C/CD.

5 EXAMPLE OF USE

We give an example of the code required to solve a set of equations using the MA43 package. To illustrate the full calling sequence for MA43, we solve for one right-hand side at the time of the factorization and then solve $\mathbf{A}^{\mathrm{T}}\mathbf{x}=\mathbf{b}$ for two right-hand sides.

Consider the 4×4 matrix

$$\begin{pmatrix} 3 & 2 & 0 & 0 \\ 1 & 4 & 3 & 0 \\ 0 & 1 & 5 & 4 \\ 0 & 0 & 2 & 1 \end{pmatrix} .$$

We want to solve $\mathbf{Ax}=\mathbf{b}$ for the right-hand side

$$\begin{pmatrix} 5 \\ 8 \\ 10 \\ 3 \end{pmatrix}$$

and to solve $\mathbf{A}^{\mathrm{T}}\mathbf{x}=\mathbf{b}$ for the two right-hand sides

$$\begin{pmatrix} 4 \\ 7 \\ 10 \\ 5 \end{pmatrix} \quad \text{and} \quad \begin{pmatrix} 1 \\ 3 \\ -2 \\ -4 \end{pmatrix}.$$

The following program is used to solve this problem.

```
C Example to illustrate MA43
C
C     .. Parameters ..
      INTEGER MAXN,MAXNZ,LRHS,LMAX,LIMAX
      PARAMETER (MAXN=4,MAXNZ=16,LRHS=2,LMAX=100,LIMAX=80)
C     ..
C     .. Local Scalars ..
      INTEGER I,J,LFACT,LIFACT,LX,N,NE,NRHS
      LOGICAL MORE,TRANS
C     ..
C     .. Local Arrays ..
      DOUBLE PRECISION A(MAXNZ),B(MAXN,LRHS),CNTL(2),FACT(LMAX),
     +                 RINFO(2),X(MAXN,LRHS)
      INTEGER ICNTL(3),IFACT(LIMAX),INFO(9),IRN(MAXNZ),IW(MAXN+1),
     +        JCN(MAXNZ),KEEP(2*MAXN+60)
C     ..
C     .. External Subroutines ..
      EXTERNAL MA43AD,MA43BD,MA43CD,MA43ID
C     ..
C Input the matrix
      READ (5,FMT=*) N,NE
      DO 10 I = 1,NE
         READ (5,FMT=*) IRN(I),JCN(I),A(I)
   10 CONTINUE
C Enter right-hand sides
      READ (5,FMT=*) NRHS
      DO 20 J = 1,NRHS
         READ (5,FMT=*) (B(I,J),I=1,N)
   20 CONTINUE
C
C Call MA43I/ID
      CALL MA43ID(ICNTL,CNTL)
C Call MA43A/AD
      MORE = .TRUE.
      CALL MA43AD(N,NE,MORE,IRN,JCN,A,IW,KEEP,ICNTL,INFO)
C Trap fatal errors.
      IF (INFO(1).LT.0) THEN
         WRITE (6,FMT=*) ' Unexpected error from MA43A/AD'
         GO TO 60
      END IF
C Call MA43B/BD
      LFACT = INFO(2)+INFO(3)*NRHS
      LIFACT = INFO(4)
      LX = MAXN
      CALL MA43BD(N,NE,JCN,A,NRHS,LX,B,X,LFACT,FACT,LIFACT,IFACT,
     +            KEEP,ICNTL,CNTL,INFO,RINFO)
```

```
      IF (INFO(1).LT.0) THEN
         WRITE (6,FMT=*) ' Unexpected error from MA43B/BD'
         GO TO 60
      END IF
      WRITE (6,FMT=9020) (INFO(J),J=1,9)
      WRITE (6,FMT=9060)
      DO 30 J = 1,NRHS
         WRITE (6,FMT=9030) J
         WRITE (6,FMT=9040) (X(I,J),I=1,N)
   30 CONTINUE
C
C Enter right-hand sides for transpose system.
      READ (5,FMT=*) NRHS
      DO 40 J = 1,NRHS
         READ (5,FMT=*) (B(I,J),I=1,N)
   40 CONTINUE
C Call MA43C/CD
      TRANS = .TRUE.
      CALL MA43CD(TRANS,N,NRHS,LX,B,X,LFACT,FACT,LIFACT,IFACT,
     +            KEEP,ICNTL,INFO)
      IF (INFO(1).LT.0) THEN
         WRITE (6,FMT=*) ' Unexpected error from MA43C/CD'
         GO TO 60
      END IF
      WRITE (6,FMT=9050)
      DO 50 J = 1,NRHS
         WRITE (6,FMT=9030) J
         WRITE (6,FMT=9040) (X(I,J),I=1,N)
   50 CONTINUE
      GO TO 70
C
   60 WRITE (6,FMT=9000)
      WRITE (6,FMT=9010) INFO(1)
   70 CONTINUE
      STOP
 9000 FORMAT (/3X,'*** Error return ***')
 9010 FORMAT (3X,'INFO(1) = ',I3)
 9020 FORMAT (' INFO   = ',9I4,/)
 9030 FORMAT (/3X,'The solution for right-hand side number',I2,
     +        ' is:')
 9040 FORMAT (4F12.4)
 9050 FORMAT (/3X,'*** A(T)x = b ***')
 9060 FORMAT (/3X,'*** Ax = b ***')
      END
```

The input data used for this problem is:

```
4   10
1    1    3.0
2    2    4.0
1    2    2.0
4    3    2.0
4    4    1.0
2    3    3.0
3    2    1.0
3    4    4.0
2    1    1.0
3    3    5.0
1
5.0   8.0  10.0   3.0
2
4.0   7.0  10.0   5.0
-2.0   1.0  0.0  -3.0
```

This produces the following output:

```
 INFO   =    0  25   8  44   0   0  -1   0   0


   *** Ax = b ***

   The solution for right-hand side number 1 is:
      1.0000      1.0000      1.0000      1.0000

   *** A(T)x = b ***

   The solution for right-hand side number 1 is:
      1.0000      1.0000      1.0000      1.0000

   The solution for right-hand side number 2 is:
     -1.0000      1.0000     -1.0000      1.0000
```